# Copyright (c) 2011 Nokia Corporation.

QT += core declarative

TARGET = sensorgallery
VERSION = 2.0.1

CONFIG += mobility qt-components
MOBILITY += sensors

INCLUDEPATH += src

HEADERS += \
    src/readingvaluemodel.h \
    src/sensormanager.h \
    src/sensormodel.h

SOURCES += \
    src/main.cpp \
    src/readingvaluemodel.cpp \
    src/sensormanager.cpp \
    src/sensormodel.cpp

OTHER_FILES += \
    qml/common/InfoContent.qml \
    qml/common/TextLabel.qml

RESOURCES += rsc/common.qrc


symbian {
    TARGET = SensorGallery
    TARGET.UID3 = 0xE2D6119E
    ICON = sensorgallery.svg

    RESOURCES += rsc/symbian.qrc

    OTHER_FILES += \
        qml/symbian/main.qml \
        qml/symbian/SensorListPage.qml \
        qml/symbian/SensorReadingPage.qml \
        qml/symbian/SymbianConstants.js

    qmlfiles.sources = qml
    DEPLOYMENT += qmlfiles
}

unix:!symbian:!maemo5 {
    # Harmattan
    message(Harmattan build)
    DEFINES += Q_WS_MAEMO_6

    RESOURCES += rsc/harmattan.qrc

    OTHER_FILES += \
        qml/harmattan/main.qml \
        qml/harmattan/SensorListPage.qml \
        qml/harmattan/SensorReadingPage.qml \
        qml/harmattan/HarmattanConstants.js

    target.path = /opt/usr/bin
    desktopfile.files = qtc_packaging/debian_harmattan/$${TARGET}.desktop
    desktopfile.path = /usr/share/applications
    icon.files = sensorgallery.png
    icon.path = /usr/share/icons/hicolor/64x64/apps

    INSTALLS += target desktopfile icon
}

# End of file.
